

Addendum to the EORTC QLQ-C30 Scoring Manual: Scoring of the EORTC QLQ-C15-PAL

Mogens Groenvold and Morten Aagaard Petersen on behalf of the EORTC Quality of Life Group

July 2006

Email addresses

Karen West, the EORTC Quality of Life Unit: Karen.West@eortc.be

Mogens Groenvold: mg02@bbh.hosp.dk

Morten Aagaard Petersen map01@bbh.hosp.dk

Introduction

The EORTC QLQ-C15-PAL is an abbreviated 15-item version of the EORTC QLQ-C30 (version 3.0) developed for palliative care.

Two methods were used for developing the EORTC QLQ-C15-PAL:

1. Shortening multi-item scales by means of item response theory (IRT) and developing algorithms for scoring the shortened scales on the original response scale metric.
2. Interviews with patients and health care professionals to identify scales and/or single items that were inappropriate or not highly relevant, and therefore could be deleted.

The questionnaire and its development are described in the references (1-4). We also refer to the EORTC QLQ-C30 Scoring Manual (5).

The EORTC QLQ-C15-PAL was discussed by Echteld et al. in Palliative Medicine (6). Our response was published in same journal (7).

Target group

The QLQ-C15-PAL is recommended for use in patients with advanced, incurable, and symptomatic cancer with a median life expectancy of a few months. It is not recommended for patients receiving palliative, anti-cancer treatments including chemotherapy, radiotherapy, endocrine treatments, or palliative surgery. These patients generally have a better prognosis and are able to complete the EORTC QLQ-C30 as demonstrated in numerous clinical trials.

The EORTC QLQ-C15-PAL

The items and scales of the EORTC QLQ-C15-PAL as compared to the EORTC QLQ-C30 version 3.0 are showed in Table 1. Note that because there are fewer items in the QLQ-C15-PAL than in the QLQ-C30, the item numbers are different (e.g., item 1 in QLQ-C15-PAL is the same as item 3 in the QLQ-C30).

Scoring

Unchanged scale/items

The pain scale and the dyspnoea, insomnia, appetite loss, and constipation items are unchanged, compared to the QLQ-C30, and are therefore scored as in the EORTC QLQ-C30 Scoring Manual (5).

Quality of life scale/item

The quality of life item of the EORTC QLQ-C15-PAL is scored as a *single item* following the procedures of the EORTC QLQ-C30 Scoring Manual:

$$\text{Score} = ((\text{Raw score} - 1)/6)*100$$

The resulting EORTC QLQ-C15-PAL 'QL item' score should be compared against the corresponding item (item 30) score in the QLQ-C30, not the QLQ-C30 QL2 scale score (which is estimated from two items).

Shortened scales

The QLQ-C15-PAL scores for the four abbreviated scales measuring physical functioning, emotional functioning, nausea and vomiting, and fatigue are estimated using the algorithms in Table 2 (2, 3). Note that these four scales cannot be scored if any of their items are missing.

The scoring of the QLQ-C15-PAL can be performed using the SAS program enclosed.

REFERENCES

1. Petersen MA, Groenvold M, Bjorner JB, Aaronson NK, Conroy T, Cull A, Fayers PM, Hjermstad MJ, Sprangers MAG, Sullivan M, for the European Organisation for Research and Treatment of Cancer Quality of Life Group. Use of differential item functioning analysis to validate translations of a health status questionnaire. *Quality of Life Research* 2003; 12: 373-85.
2. Bjorner JB, Petersen MA, Groenvold M, Aaronson NK, Ahlner-Elmqvist M, Arraras JI, Fayers P, Jordhoy M, Sprangers MAG, Young T for the European Organisation for Research and Treatment of Cancer Quality of Life Group. Use of item response theory to develop a shortened version of the EORTC QLQ-C30 emotional function scale. *Quality of Life Research* 2004; 13: 1683-1697.
3. Petersen MAa, Groenvold M, Aaronson NK, Blazeby J, Brandberg Y, de Graeff A, Fayers P, Hammerlid E, Sprangers MA, Velikova G, Bjorner JB for the European Organisation for Research and Treatment of Cancer Quality of Life Group. Item response theory was used to shorten EORTC QLQ-C30 scales for use in palliative care. *Journal of Clinical Epidemiology* 2006; 59: 36-44.
4. Groenvold M, Petersen MAa, Aaronson NK, Arraras JI, Blazeby JM, Bottomley A, Fayers PM, de Graeff A, Hammerlid E, Kaasa S, Sprangers MAG, Bjorner JB for the EORTC Quality of Life Group. The development of the EORTC QLQ-C15-PAL: a shortened questionnaire for cancer patients in palliative care. *European Journal of Cancer* 2006; 42: 55-64.
5. Fayers PM, Aaronson N, Bjordal K, Groenvold M, Curran D, Bottomley A. The EORTC QLQ-C30 Scoring Manual. 3rd Ed. Brussels: European Organization for Research and Treatment of Cancer; 2001.
6. Echteld MA, Deliens L, Ontwuteaka-Philipsen B, Klein M, Wal G. EORTC QLQ-C15-PAL: the new standard in the assessment of health-related quality of life in advanced cancer? *Palliative Medicine* 2006; 20 (1): 1-2.
7. Groenvold M, Petersen MAa, Aaronson NK, Arraras JI, Blazeby JM, Bottomley A, Fayers PM, de Graeff A, Hammerlid E, Kaasa S, Sprangers MAG, Bjorner JB. EORTC QLQ-C15-PAL: the new standard in the assessment of health-related quality of life in advanced cancer? A response to Echteld MA et al, *Palliative Medicine* 2006; 20: 1-2. *Palliative Medicine* 2006; 20 (2) 59-61.

Table 1. The items and scales of the EORTC QLQ-C30 and the EORTC QLQ-C15-PAL (4).

	QLQ-C30 version 3.0			QLQ-C15-PAL		
	Scale	Number of items	Item numbers	Scale	Number of items	Item numbers
Global health status/quality of life						
Global health status/quality of life	QL2	2	29, 30	'QL item'	1	15
Functional scales						
Physical functioning	PF2	5	1-5	PF2*	3	1-3
Role functioning	RF2	2	6, 7	-	0	-
Emotional functioning	EF	4	21-24	EF*	2	13, 14
Cognitive functioning	CF	2	20, 25	-	0	-
Social functioning	SF	2	26, 27	-	0	-
Symptom scales						
Fatigue	FA	3	10, 12, 18	FA*	2	7, 11
Nausea and vomiting	NV	2	14, 15	NV*	1	9
Pain	PA	2	9, 19	PA	2	5, 12
Dyspnoea	DY	1	8	DY	1	4
Insomnia	SL	1	11	SL	1	6
Appetite loss	AP	1	13	AP	1	8
Constipation	CO	1	16	CO	1	10
Diarrhoea	DI	1	17	-	0	-
Financial difficulties	FI	1	28	-	0	-

*) These scores are predicted from the items of the shortened scales and are directly comparable with the corresponding QLQ-C30 scale scores, thus, these scale names are unchanged.

Table 2. Scoring algorithms for predicting the scores on the PF, FA, NV, and EF scales from the items of the QLQ-C15-PAL (2, 3).

Physical functioning		Fatigue			Nausea/vomiting		Emotional functioning		
Sum of items 1-3 ^a	Predicted PF2	Responses to Item 7	Item 11	Predicted FA	Response to item 9	Predicted NV	Responses to Item 13	Item 14	Predicted EF
0	0.0	‘Not at all’	‘Not at all’	0.0	‘Not at all’	0.0	‘Not at all’	‘Not at all’	100.0
1	6.7	‘Not at all’	‘A little’	22.2	‘A little’	16.7	‘Not at all’	‘A little’	83.3
2	13.3	‘Not at all’	‘Quite a bit’	33.3	‘Quite a bit’	50.0	‘Not at all’	‘Quite a bit’	66.7
3	20.0	‘Not at all’	‘Very much’	55.6	‘Very much’	100.0	‘Not at all’	‘Very much’	50.0
4	26.7	‘A little’	‘Not at all’	22.2			‘A little’	‘Not at all’	83.3
5	33.3	‘A little’	‘A little’	33.3			‘A little’	‘A little’	66.7
6	46.7	‘A little’	‘Quite a bit’	55.6			‘A little’	‘Quite a bit’	50.0
7	60.0	‘A little’	‘Very much’	66.7			‘A little’	‘Very much’	41.7
8	73.3	‘Quite a bit’	‘Not at all’	33.3			‘Quite a bit’	‘Not at all’	66.7
9	93.3	‘Quite a bit’	‘A little’	44.4			‘Quite a bit’	‘A little’	50.0
		‘Quite a bit’	‘Quite a bit’	66.7			‘Quite a bit’	‘Quite a bit’	41.7
		‘Quite a bit’	‘Very much’	88.9			‘Quite a bit’	‘Very much’	16.7
		‘Very much’	‘Not at all’	44.4			‘Very much’	‘Not at all’	50.0
		‘Very much’	‘A little’	66.7			‘Very much’	‘A little’	41.7
		‘Very much’	‘Quite a bit’	88.9			‘Very much’	‘Quite a bit’	16.7
		‘Very much’	‘Very much’	100.0			‘Very much’	‘Very much’	0.0

^a: Items 1-3 are scored: 0 = ‘Very much’, 1 = ‘Quite a bit’, 2 = ‘A little’, 3 = ‘Not at all’.

SAS program for scoring the EORTC QLQ-C15-PAL

```
/*
> INSTALLATION AND HOW TO USE
To run the commands, first load your data set into SAS. Then copy this syntax to
the SAS Program Editor and insert the following line at the end of the file:
  %palscore(mydata);
  where mydata is the name of your data set. Then run this SAS syntax (including
your added line).

> ITEM NAMES
The QLQ-C15-PAL items should be named q1, q2, q3, ... q15.
Q1-q14 should be scored 1-4 with 1='Not at all', 2='A little', 3='Quite a bit',
and 4='Very much'.
Q15 should be scored 1-7 with 1='Very poor' and 7='Excellent'.

> WHAT YOU GET
Function scales, symptom scales and single-item scales are calculated and
labeled in accordance with the descriptions for the QLQ-C15-PAL. The shortened
scales PF, EF, FA, and NV are scored using the special QLQ-C15-PAL scoring while
the remaining scales are scored according to the original scoring procedures
described in the QLQ-C30 Scoring Manual. For example, the emotional functioning
scale is calculated from items q21 and q24. It is called EF, and is labeled
"Emotional functioning". It is scaled to range from 0 to 100, with a high score
indicating high emotional functioning. Note that if you have an existing
variable in your data set called EF, it will be automatically over-written. If
this is not what you want, you should copy the original data set and store it
under a different name for this application: your revised data set will
automatically be copied into your original SAS data set.

> OUT-OF-RANGE VALUES
This program checks for items with out-of-range values and treats them as
missing.

> MISSING ITEMS
The default action of this program for the scales (PA) scored according to the
original QLQ-C30 scoring procedures is to impute missing items provided at least
half the scale items are present; the method is described in the QLQ-C30 Scoring
Manual. For the four shortened scales if one or more items is missing the scale
is set to missing.
*/

/* SAS COMMANDS to generate QLQ-C15-PAL scales. */
/*
=====
*/
/* Morten Aagaard Petersen and Mogens Groenvold, 26-8-2005
   Adapted from Kristel Van Steen and Desmond Curran, 30-12-1998 Amended PF 7-
2002 */
/*
=====
*/
/* This code is provided as is without warranty of any kind, either */
/* expressed or implied. It is not guaranteed to be error free, and in */
/* no event will we accept liability for any damages arising out of the */
/* use of this code. */
```

```

/*=====*/
/*****/
/* Cleaning the Data */
/*****/
%macro clean(data);
proc format;
value      item4_ 1='Not at All'
              2='A Little'
              3='Quite a Bit'
              4='Very Much';
value      item7_ 1='Very Poor'
              7='Excellent';

data &data;
  set &data;
  %do I=1 %to 14;
    if (q&i<1 or q&i>4) then q&i=.;
    format q&i item4_.;
  %end;
  if (q15<1 or q15>7) then q15=.;
  format q&i item7_.;
run;
%mend clean;

/*****/
/* Definitions for Score Calculation */
/*****/
%macro score(data,type,scale,items);
%let i=1;
  %do %while(%length(%scan(&items,&i)) ^= 0);
    %let i=%eval(&i+1);
  %end;
%let i=%eval(&i-1) ;
data &data;
  set &data;
  XMEAN = MEAN(OF &items);
  XNUM = N (of &items);
  %if (%upcase(&type)=S) %then %do ;
    If (XNUM*2 GE &i) THEN &scale = ((XMEAN-1)/3)*100;
  %end ;
  %if (%upcase(&type)=G) %then %do ;
    If (XNUM*2 GE &i) THEN &scale = (((XMEAN-1)/6)*100);
  %end ;
run;
%mend score;

/*****/
/* Returning a Data Set with Calculated Scores for Each Scale:*/
/* Basic Macro */
/*****/
%macro palscore(data);
  %clean(&data);
/*****Scoring of original QLQ-C30 scales/single-item measures*****/
  %score(data=&data , type=s, scale=pa, items= q5 q12);
  %score(data=&data , type=s, scale=dy, items= q4);
  %score(data=&data , type=s, scale=sl, items= q6);
  %score(data=&data , type=s, scale=ap, items= q8);
  %score(data=&data , type=s, scale=co, items= q10);
  %score(data=&data , type=g, scale=ql2, items= q15);

```

```

Data &data;
set &data ;
/*****Scoring of the shortened scales*****/
*PF2;
pfsum=12-(q1+q2+q3);
if 0<=pfsum<=5 then pf2=pfsum*100/15;
if pfsum=6 then pf2=7*100/15;
if pfsum=7 then pf2=9*100/15;
if pfsum=8 then pf2=11*100/15;
if pfsum=9 then pf2=14*100/15;

*FA;
fasum=q7+q11-2;
if fasum=0 then fa=0;
if fasum=1 then fa=2*100/9;
if fasum=2 then fa=3*100/9;
if (q7=3 and q11=2) or (q7=4 and q11=1) then fa=4*100/9;
if (q7=1 and q11=4) or (q7=2 and q11=3) then fa=5*100/9;
if fasum=4 then fa=6*100/9;
if fasum=5 then fa=8*100/9;
if fasum=6 then fa=100;

*NV;
if q9=1 then nv=0;
if q9=2 then nv=100/6;
if q9=3 then nv=50;
if q9=4 then nv=100;

*EF;
efsum=8-(q13+q14);
if efsum=0 then ef=0;
if efsum=1 then ef=2*100/12;
if efsum=2 then ef=5*100/12;
if efsum=3 then ef=50;
if efsum=4 then ef=8*100/12;
if efsum=5 then ef=10*100/12;
if efsum=6 then ef=100;

LABEL q1='SHORT WALK';
LABEL q2='IN BED';
LABEL q3='NEED HELP';
LABEL q4='SHORT OF BREATH';
LABEL q5='PAIN';
LABEL q6='TROUBLE SLEEPING';
LABEL q7='FELT WEAK';
LABEL q8='LACKED APPETITE';
LABEL q9='FELT NAUSEATED';
LABEL q10='BEEN CONSTIPATED';
LABEL q11='BEEN TIRED';
LABEL q12='PAIN INTERFERENCE';
LABEL q13='FELT TENSE';
LABEL q14='FELT DEPRESSED';
LABEL q15='QUALITY OF LIFE';

LABEL pf2='Physical Functioning';
LABEL ef='Emotional Functioning';
LABEL q12='Overall quality of life (q30)';
LABEL fa='Fatigue';
LABEL nv='Nausea / Vomiting';

```

```
LABEL pa='Pain';  
LABEL dy='Dyspnoea';  
LABEL sl='Insomnia';  
LABEL ap='Appetite loss';  
LABEL co='Constipation';  
drop xnum xmean pfsun fasum efsun;  
run ;
```

```
%mend palscore;
```